MATH20C - Final - Allen

Problem 1 (10 points)

Evaluate the quadruple integral
$\int_0^2 \int_0^1 \int_0^{x+y} \int_y^x \, dt\, dz\, dy\, dx$

Solution

$\int_0^2 \int_0^1 \int_0^{x+y} \int_y^x \, dt\, dz\, dy\, dx$
$= \int_0^2 \int_0^1 \int_0^{x+y} (x-y)\, dz\, dy\, dx$
$= \int_0^2 \int_0^1 (x^2-y^2)\, dy\, dx$
$= \int_0^2 (x^2-\frac{1}{3})\, dx$
$= 2$

```
Integrate[1, {x, 0, 2}, {y, 0, 1}, {z, 0, x + y}, {t, y, x}]
```

```
2
```

Problem 2 (10 points)

The area of a region is given by a sum of double integrals:
Area= $\int_0^1 \int_{\sqrt{1-x^2}}^{\sqrt{1-x^2/2}} dy\, dx + \int_1^{\sqrt{2}} \int_0^{\sqrt{1-x^2/2}} dy\, dx$

Simplify the expression for area by reversing the order of integration and finding the appropriate limits:
Area= $\int_{y_0}^{y_1} \int_{x_0(y)}^{x_1(y)} dx\, dy$

You do not need to evaluate the integral. Hint: the inner curve is given by $x^2 + y^2 = 1$, while the outer curve is given by $\frac{x^2}{2} + y^2 = 1$

Solution

```
CylindricalDecomposition[x^2 + y^2 > 1 && x^2/2 + y^2 < 1 && x > 0 && y > 0, {y, x}]
```

$0 < y < 1 \;\&\&\; \sqrt{1-y^2} < x < \sqrt{2-2y^2}$

area = $\int_0^1 \int_{\sqrt{1-y^2}}^{\sqrt{2-2y^2}} dx\, dy$

Just for fun (not necessary to receive full credit):

```
Integrate[1, {y, 0, 1}, {x, Sqrt[1 - y^2], Sqrt[2 - 2 y^2]}]
```

$\frac{1}{4}\left(-1+\sqrt{2}\right)\pi$

1/4 the area of an ellipse (π a b, with a/b the semi-major/minor axes) minus 1/4 the area of a disk ($\pi\, r^2$):

$\frac{1}{4}\pi\sqrt{2} - \frac{1}{4}\pi\, 1^2$ **// Simplify**

$\frac{1}{4}\left(-1+\sqrt{2}\right)\pi$

Problem 3 (10 points)

Find the point (x_0, y_0) such that the following function is locally extremized:
f(x,y) = $x^2 + 4\,x\,y - 10\,x - y^2 + 5$,
and determine whether it corresponds to a max, min, or saddle point.

Solution

$\mathbf{f} = \frac{1}{2}\{x-1, y-2\}.\begin{pmatrix} 2 & 4 \\ 4 & -2 \end{pmatrix}.\{x-1, y-2\}$ **// Expand**

$\mathbf{Solve}[\{\partial_x f, \partial_y f\} == \{0, 0\}]$

$5 - 10\,x + x^2 + 4\,x\,y - y^2$

$\{\{x \to 1, y \to 2\}\}$

$\mathbf{H} = \mathbf{Table}[\partial_i \partial_j f, \{i, \{x, y\}\}, \{j, \{x, y\}\}];$
H // MatrixForm

$\begin{pmatrix} 2 & 4 \\ 4 & -2 \end{pmatrix}$

H // Det

-20

Saddle point at (x_0, y_0) = (1, 2)

Problem 4 (10 points)

Write an expression of the form, ax+ by + c =0, for the tangent line at the point (x_0, y_0)=(2,1) for the implicit curve:
$x^3 - \frac{9\,x\,y}{2} + y^3 = 0$

Solution

Take the gradient of the constraint and dot into $(x - x_0,\ y - y_0)$:

```
f = x^3 + y^3 - 9/2 x y;
n = {∂x f, ∂y f} /. {x → 2, y → 1};
n.{x - 2, y - 1} // Expand
```

$$-9 + \frac{15\,x}{2} - 6\,y$$

Problem 5 (10 points)

A path you might trace out along a carousel can be parametrized as follows:

$\vec{r}\,(t)\ =\ (2\cos(2\,\pi\,t),\ \ 2\sin(2\,\pi\,t),\ \ \sin(10\,\pi\,t)\,).$

Find parametric equations for the tangent line, $\vec{T}$(t), at the point t_0= 1/4

Solution

```
Clear[r];
t0 = 1 / 4;
r[t_] = {2 Cos[2 π t], 2 Sin[2 π t], Sin[10 π t]};
T[t_] = r[t0] + t r'[t0]
```

```
{-4 π t, 2, 1}
```

Problem 6 (10 points)

Evaluate the following derivatives,

$$\left.\frac{\partial f}{\partial u}\right)_v \text{ and } \left.\frac{\partial f}{\partial v}\right)_u,$$

at $(x_0,\ y_0,\ z_0) = (1{,}0{,}2)$, and $(u_0,\ v_0) = (0,\ -1)$, given the following data, $f(x,\ y,\ z) = x^2 + y\,z$, and $x(u,\ v) =$ $u + 1$, $\ y\,(u,\ v) =$ u v, z(u,v) = -v+1.

Solution

```
f = x^2 + y z;
r = {u + 1, u v, -v + 1};
gradf = {∂x f, ∂y f, ∂z f} /. {x → 1, y → 0, z → 2}
drdu = ∂u r /. {u → 0, v → -1}
drdv = ∂v r /. {u → 0, v → -1}
```

{2, 2, 0}

{1, -1, 0}

{0, 0, -1}

```
gradf.drdu
gradf.drdv
```

0

0

Problem 7 (10 points)

Write an expression of the form, a x + b y + c z + d =0, for the tangent plane at the point (x_0, y_0, z_0)=(1,1,1), for the following implicit surface:
f(x,y,z) = $x\,y\,z = 1$

Solution

The point lives on the surface:

```
{x0, y0, z0} = {1, 1, 1};
x y z /. {x → x0, y → y0, z → z0}
```

1

```
f = x y z;
n = {∂x f, ∂y f, ∂z f} /. {x → x0, y → y0, z → z0};
n.{x - x0, y - y0, z - z0} // Expand
```

$-3 + x + y + z$

Problem 8 (10 points)

Find the extrema of f, given the constraint g:
f(x,y) = $-x^2 - 2\,x\,y + x - y^2 + 3\,y$, g(x,y) = $y - x = -1$

Solution

```
f = -x^2 - 2 x y + x - y^2 + 3 y;
g = y - x;
{∂x f == λ ∂x g, ∂y f == λ ∂y g, g == -1}
Solve[%, {x, y, λ}]
```

$\{1 - 2x - 2y == -\lambda,\ 3 - 2x - 2y == \lambda,\ -x + y == -1\}$

$\{\{x \to 1,\ y \to 0,\ \lambda \to 1\}\}$

We only care about (x_0, y_0) so $(x_0, y_0) = (1, 0)$

Problem 9 (10 points)

Find all two extrema of f, given two constraints g_1 and g_2:

f(x,y) = z

$g_1(x, y) = x y = 1$

$g_2(x, y) = x + y + z = 4$

Solution

```
f = z;
g1 = x y;
g2 = x + y + z;
{∂x f == λ1 ∂x g1 + λ2 ∂x g2, ∂y f == λ1 ∂y g1 + λ2 ∂y g2,
 ∂z f == λ1 ∂z g1 + λ2 ∂z g2, g1 == 1, g2 == 4}
Solve[%, {x, y, z, λ1, λ2}]
```

$\{0 == y\,\lambda 1 + \lambda 2,\ 0 == x\,\lambda 1 + \lambda 2,\ 1 == \lambda 2,\ x\,y == 1,\ x + y + z == 4\}$

$\{\{x \to -1,\ y \to -1,\ z \to 6,\ \lambda 1 \to 1,\ \lambda 2 \to 1\},\ \{x \to 1,\ y \to 1,\ z \to 2,\ \lambda 1 \to -1,\ \lambda 2 \to 1\}\}$

We only care about (x_0, y_0, z_0) so $(x_0, y_0, z_0) = (-1, -1, 6)$, $(1, 1, 2)$